

# NAG Fortran Library Routine Document

## F08NSF (CGEHRD/ZGEHRD)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08NSF (CGEHRD/ZGEHRD) reduces a complex general matrix to Hessenberg form.

### 2 Specification

```
SUBROUTINE F08NSF(N, ILO, IHI, A, LDA, TAU, WORK, LWORK, INFO)
ENTRY      cgehrd (N, ILO, IHI, A, LDA, TAU, WORK, LWORK, INFO)
INTEGER    N, ILO, IHI, LDA, LWORK, INFO
complex  A(LDA,*), TAU(*), WORK(*)
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine reduces a complex general matrix  $A$  to upper Hessenberg form  $H$  by a unitary similarity transformation:  $A = QHQ^H$ .  $H$  has real subdiagonal elements.

The matrix  $Q$  is not formed explicitly, but is represented as a product of elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with  $Q$  in this representation (see Section 8).

The routine can take advantage of a previous call to F08NVF (CGEBAL/ZGEBAL), which may produce a matrix with the structure:

$$\begin{pmatrix} A_{11} & A_{12} & A_{13} \\ & A_{22} & A_{23} \\ & & A_{33} \end{pmatrix}$$

where  $A_{11}$  and  $A_{33}$  are upper triangular. If so, only the central diagonal block  $A_{22}$ , in rows and columns  $i_{lo}$  to  $i_{hi}$ , needs to be reduced to Hessenberg form (the blocks  $A_{12}$  and  $A_{23}$  will also be affected by the reduction). Therefore the values of  $i_{lo}$  and  $i_{hi}$  determined by F08NVF (CGEBAL/ZGEBAL) can be supplied to the routine directly. If F08NVF (CGEBAL/ZGEBAL) has not previously been called however, then  $i_{lo}$  must be set to 1 and  $i_{hi}$  to  $n$ .

### 4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

1: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

2: ILO – INTEGER *Input*

3: IHI – INTEGER *Input*

*On entry:* if  $A$  has been output by F08NVF (CGEBAL/ZGEBAL), then ILO and IHI **must** contain the values returned by that routine. Otherwise, ILO must be set to 1 and IHI to  $N$ .

*Constraints:*

$$1 \leq \text{ILO} \leq \text{IHI} \leq N \text{ if } N > 0, \\ \text{ILO} = 1 \text{ and } \text{IHI} = 0 \text{ if } N = 0.$$

- 4: A(LDA,\*) – **complex** array *Input/Output*  
**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  general matrix  $A$ .  
*On exit:*  $A$  is overwritten by the upper Hessenberg matrix  $H$  and details of the unitary matrix  $Q$ . The subdiagonal elements of  $H$  are real.
- 5: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F08NSF (CGEHRD/ZGEHRD) is called.  
*Constraint:*  $\text{LDA} \geq \max(1, N)$ .
- 6: TAU(\*) – **complex** array *Output*  
**Note:** the dimension of the array TAU must be at least  $\max(1, N - 1)$ .  
*On exit:* further details of the unitary matrix  $Q$ .
- 7: WORK(\*) – **complex** array *Workspace*  
**Note:** the dimension of the array WORK must be at least  $\max(1, \text{LWORK})$ .  
*On exit:* if  $\text{INFO} = 0$ , the real part of  $\text{WORK}(1)$  contains the minimum value of  $\text{LWORK}$  required for optimum performance.
- 8: LWORK – INTEGER *Input*  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08NSF (CGEHRD/ZGEHRD) is called, unless  $\text{LWORK} = -1$ , in which case a workspace query is assumed and the routine only calculates the optimal dimension of WORK (using the formula given below).  
*Suggested value:* for optimum performance  $\text{LWORK}$  should be at least  $N \times nb$ , where  $nb$  is the **blocksize**.  
*Constraint:*  $\text{LWORK} \geq \max(1, N)$  or  $\text{LWORK} = -1$ .
- 9: INFO – INTEGER *Output*  
*On exit:*  $\text{INFO} = 0$  unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$\text{INFO} < 0$

If  $\text{INFO} = -i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed Hessenberg matrix  $H$  is exactly similar to a nearby matrix  $A + E$ , where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$  is a modestly increasing function of  $n$ , and  $\epsilon$  is the **machine precision**.

The elements of  $H$  themselves may be sensitive to small perturbations in  $A$  or to rounding errors in the computation, but this does not affect the stability of the eigenvalues, eigenvectors or Schur factorization.

## 8 Further Comments

The total number of real floating-point operations is approximately  $\frac{8}{3}q^2(2q + 3n)$ , where  $q = i_{hi} - i_{lo}$ ; if  $i_{lo} = 1$  and  $i_{hi} = n$ , the number is approximately  $\frac{40}{3}n^3$ .

To form the unitary matrix  $Q$  this routine may be followed by a call to F08NTF (CUNGHR/ZUNGHR):

```
CALL CUNGHR (N, ILO, IHI, A, LDA, TAU, WORK, LWORK, INFO)
```

To apply  $Q$  to an  $m$  by  $n$  complex matrix  $C$  this routine may be followed by a call to F08NUF (CUNMHR/ZUNMHR). For example,

```
CALL CUNMHR ('Left', 'No Transpose', M, N, ILO, IHI, A, LDA, TAU, C, LDC,
+          WORK, LWORK, INFO)
```

forms the matrix product  $QC$ .

The real analogue of this routine is F08NEF (SGEHRD/DGEHRD).

## 9 Example

To compute the upper Hessenberg form of the matrix  $A$ , where

$$A = \begin{pmatrix} -3.97 - 5.04i & -4.11 + 3.70i & -0.34 + 1.01i & 1.29 - 0.86i \\ 0.34 - 1.50i & 1.52 - 0.43i & 1.88 - 5.38i & 3.36 + 0.65i \\ 3.31 - 3.85i & 2.50 + 3.45i & 0.88 - 1.08i & 0.64 - 1.48i \\ -1.10 + 0.82i & 1.81 - 1.59i & 3.25 + 1.33i & 1.57 - 3.44i \end{pmatrix}.$$

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08NSF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5, NOUT=6)
INTEGER          NMAX, LDA, LWORK
PARAMETER       (NMAX=8, LDA=NMAX, LWORK=64*NMAX)
complex        ZERO
PARAMETER       (ZERO=(0.0e0, 0.0e0))
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
*      .. Local Arrays ..
complex        A(LDA, NMAX), TAU(NMAX-1), WORK(LWORK)
CHARACTER       CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL        X04DBF, cgehrd
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08NSF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
READ (NIN,*) ((A(I,J), J=1,N), I=1,N)
*
*      Reduce A to upper Hessenberg form
*
CALL cgehrd(N, 1, N, A, LDA, TAU, WORK, LWORK, INFO)
*

```

```

*      Set the elements below the first sub-diagonal to zero
*
      DO 40 I = 1, N - 2
        DO 20 J = I + 2, N
          A(J,I) = ZERO
20      CONTINUE
40      CONTINUE
*
*      Print upper Hessenberg form
*
      WRITE (NOUT,*)
      IFAIL = 0
*
      CALL X04DBF('General',' ',N,N,A,LDA,'Bracketed','F7.4',
+              'Upper Hessenberg form','Integer',RLABS,'Integer',
+              CLABS,80,0,IFAIL)
*
      END IF
      STOP
      END

```

## 9.2 Program Data

F08NSF Example Program Data

```

4
(-3.97,-5.04) (-4.11, 3.70) (-0.34, 1.01) ( 1.29,-0.86) :Value of N
( 0.34,-1.50) ( 1.52,-0.43) ( 1.88,-5.38) ( 3.36, 0.65)
( 3.31,-3.85) ( 2.50, 3.45) ( 0.88,-1.08) ( 0.64,-1.48)
(-1.10, 0.82) ( 1.81,-1.59) ( 3.25, 1.33) ( 1.57,-3.44) :End of matrix A

```

## 9.3 Program Results

F08NSF Example Program Results

Upper Hessenberg form

```

          1          2          3          4
1 (-3.9700,-5.0400) (-1.1318,-2.5693) (-4.6027,-0.1426) (-1.4249, 1.7330)
2 (-5.4797, 0.0000) ( 1.8585,-1.5502) ( 4.4145,-0.7638) (-0.4805,-1.1976)
3 ( 0.0000, 0.0000) ( 6.2673, 0.0000) (-0.4504,-0.0290) (-1.3467, 1.6579)
4 ( 0.0000, 0.0000) ( 0.0000, 0.0000) (-3.5000, 0.0000) ( 2.5619,-3.3708)

```

---